






# Automatic Port to OpenACC/OpenMP for Physical Parameterization in Climate and Weather Code Using the CLAW Compiler

Valentin Clement<sup>1</sup> , Philippe Marti<sup>1</sup> , Xavier Lapillonne<sup>2</sup> ,  
Oliver Fuhrer<sup>2</sup> , William Sawyer<sup>3</sup> 

© The Authors 2019. This paper is published with open access at SuperFri.org

In order to benefit from emerging high-performance computing systems, weather and climate models need to be adapted to run efficiently on different hardware architectures such as accelerators. This is a major challenge for existing community models that represent extremely large codebase written in Fortran. Large parts of the code can be ported using OpenACC compiler directives but for time-critical components such as physical parameterizations, code restructuring and optimizations specific to a hardware architecture are necessary to obtain high performance. In an effort to retain a single source code for multiple target architectures, the CLAW Compiler and the CLAW Single Column Abstraction were introduced. We report on the extension of the CLAW SCA to handle ELEMENTAL functions and subroutines. We demonstrate the new capability on the JSBACH land surface scheme of the ICON climate model. With the extension, JSBACH can be automatically ported to OpenACC or OpenMP for accelerators with minimal to no change to the original code.

*Keywords: compiler, directive, GPU, OpenACC, OpenMP, automatic port.*

## Introduction

Numerical Weather Prediction and Climate modeling can highly benefit from computer technologies advances by increasing resolution or model complexity. In the recent years, architectures such as Graphics Processing Unit (GPU) have emerged for scientific high performance computing offering new opportunities. Most of the current Numerical Weather Prediction and Climate models are large Fortran based community codes which require to be adapted or re-written to run on non traditional CPU architectures such as GPU. In order to prepare for heterogenous supercomputer architectures, the global weather and climate model ICON [2, 4] is being ported to accelerators. The major part of the porting is currently achieved using OpenACC [10] compiler directives. For many code sections simple insertion of OpenACC compiler directives is an appropriate porting approach. But, for time-critical components such as the physical parameterizations, code restructuring and optimizations it is necessary to obtain optimal performance [6].

In some cases, the required code restructuring and optimization for GPU architectures have a negative impact when running the same code on a CPU architecture. Multiple solutions to address this problem have been proposed to achieve performance portability across architectures [3, 5, 8]. Here we focus on CLAW [1], an open-source source-to-source translator that allows to perform architecture-specific code transformations with minimal code modifications.

The CLAW SCA is designed to address the physical parameterizations of atmospheric models in Fortran. Physical parameterizations are typically horizontally independent so each vertical column can be computed separately. With the CLAW SCA, the physical parameterizations are written in Fortran only considering the vertical dependencies while the horizontal directions are abstracted out. The CLAW Compiler can transform the code for a specific target architecture

<sup>1</sup>Centre for Climate System Modeling, Zurich, Switzerland

<sup>2</sup>Federal Office of Meteorology and Climatology MeteoSwiss, Zurich, Switzerland

<sup>3</sup>CSCS Swiss National Supercomputing Centre, Lugano, Switzerland

and insert horizontal loops and compiler directives such as OpenMP [11] for accelerator (version  $\geq 4.5$ ) or OpenACC.

In this paper, we focus on an CLAW SCA incorporation that extends CLAW capability to address such performance portability issues without imposing disturbing changes to the code base. We introduce a special case of the CLAW SCA connected with ELEMENTAL functions and subroutines. We demonstrate the new feature on the JSBACH land surface scheme [7] used in the ICON climate model. This paper is structured as follows: Section 1 describes the JSBACH land surface scheme and its use of ELEMENTALS. In Sections 2 and 3 we present the CLAW Compiler and the extension of SCA for ELEMENTALS. In Sections 4 and 5 we discuss our performance results and some code metrics. Finally, in Section 5 we draw our conclusion and discuss future work.

## 1. The JSBACH Land Surface Scheme and ELEMENTAL

In this section, we briefly describe the concept of ELEMENTAL subroutines and functions and its application in the JSBACH land surface scheme. Further, we summarize its structure.

### 1.1. ELEMENTAL subroutines and functions

In Fortran, an ELEMENTAL function or subroutine is defined as a scalar operator. Dummy arguments as well as potential return value must be scalars but it may be called with arrays of arbitrary dimensionality as actual arguments. In this case, the operations defined in the function or the subroutine are applied element-wise on the full arrays. The main benefit of ELEMENTAL functions or subroutines is that it allows the user to write more compact code and allows the compiler to parallelize function or subroutine execution.

```

1 PROGRAM main
2   IMPLICIT NONE
3   INTEGER :: x, y, z
4   INTEGER, DIMENSION(10) :: xa, ya, za
5   x = 2; y = 10
6   xa(:) = 2; ya(:) = 5
7
8   ! Call ELEMENTAL with scalars
9   z = power(x, y)
10  print*, 'x ** y = ', z
11
12  ! Call ELEMENTAL with arrays
13  za(:) = power(xa(:), ya(:))
14  print*, 'xa(:) ** ya(:) = ', za
15
16 CONTAINS
17
18  ELEMENTAL FUNCTION power(a, b) RESULT(c)
19    INTEGER, INTENT(IN) :: a, b
20    INTEGER :: c
21    c = a ** b
22  END FUNCTION power
23 END PROGRAM
```


**Figure 1.** Code example with a basic ELEMENTAL function

A very basic ELEMENTAL function is implemented in Fig. 1 from line 18 to 22. First it is mentioned with scalar arguments at line 9, second, with arrays arguments at line 13.

## 1.2. JSBACH

JSBACH is the land surface scheme of the global ICON model in climate mode. It is designed to serve as a land surface boundary for the atmosphere in the coupled simulation but it can also be used offline as a standalone model. JSBACH simulates photosynthesis, phenology and land physics with hydrological and bio-geochemical cycles.

JSBACH code is written in 2008 Fortran object-oriented style. It is designed as a pipeline of tasks where each task is applied on various tiles of the soil depending on their properties. Each task retrieves a number of fields from the internal memory object and applies several ELEMENTAL functions and/or subroutines as well as vector operations to them.



**Figure 2.** Example of a typical task workflow using ELEMENTALs in JSBACH - update radiation surface from the radiation interface

To hide some aspects of the design such as the memory access and pointers, JSBACH includes a non-Fortran Domain Specific Language (DSL). The code must be preprocessed by a Python script in order to obtain standard Fortran code. Section 2.1 describes where this preprocessing sits in the CLAW Compiler workflow. The choices of using ELEMENTAL functions and subroutines and the JSBACH DSL significantly simplify the code for the domain scientist.

In its original form, the code is not suited to be ported easily to accelerators using OpenACC or OpenMP as compilers do not allow directives in PURE or ELEMENTAL functions and subroutines. In some cases vector notation can be handled by compilers with the *!\$acc kernels* construct. But in practice, we have experienced several cases where the compiler was not able to determine that the kernel can be run in parallel and thus generated a sequential version of the code resulting in a significantly slower GPU execution comparing to the CPU version. This was especially true with older version of OpenACC compatible compiler but can be solved having reasonable amount of vector notation in an *!\$acc kernels* block.

In order to execute the code on a GPU, we need to either fundamentally refactor JSBACH or to take advantage of a source-to-source translator such as CLAW. CLAW can automatize the port while taking advantage of the current information we can extract from it. The latter solution is the one described in this paper and has the advantage to bring portability across compiler directives by supporting OpenACC and OpenMP simultaneously.

Figure 2 is a typical call graph of a single task defined in JSBACH. `update_radiation_surface` is part of the radiation interface. This task is calling a single `ELEMENTAL` subroutine three times with different fields as arguments. This `ELEMENTAL` subroutine (`calc_radiation_surface_net`) calls another `ELEMENTAL` function.

As all `ELEMENTAL` code needs to be transformed to accept compiler directives, everything in the task is transformed to be executed on the accelerator.

**Table 1.** JSBACH tasks with the number of automatically generated kernel and the number of 1 dimensional (1D) and 2 dimensional (2D) input/output fields for each task

Interface	Task name	Nb. kernels	1D in	1D out	2D in	2D out
Radiation	<code>surface_radiation</code>	4	11	7	0	10
	<code>radiation_par</code>	5	10	6	4	4
	<code>albedo</code>	12	38	18	0	0
Phenology	<code>phenology_logrop</code>	11	30	21	0	0
	<code>fpc</code>	3	2	1	0	10
Hydrology	<code>snow_and_skin_fraction</code>	11	11	4	0	0
	<code>soil_properties</code>	2	7	0	7	7
	<code>evaporation</code>	3	16	6	0	0
	<code>surface_hydrology</code>	1	4	0	4	0
	<code>soil_hydrology</code>	3	3	3	1	0
	<code>canopy_cond_unstressed</code>	1	2	1	0	0
	<code>canopy_cond_stressed</code>	1	7	5	5	0
	<code>water_balance</code>	5	7	3	2	3
Surface energy balance	<code>surface_energy_lake</code>	7	16	16	0	0
	<code>surface_energy_land</code>	18	23	14	0	0
	<code>surface_fluxes_lake</code>	1	11	6	0	0
	<code>surface_fluxes_land</code>	1	7	4	0	0
	<code>asselin_land</code>	3	3	2	0	0
	<code>snowmelt_correction</code>	4	3	1	4	0
Soil snow energy	<code>soil_and_snow_properties</code>	14	2	4	2	2
	<code>soil_and_snow_temperature</code>	14	3	5	4	10
Assimilation	<code>assimilation_scaling_factors</code>	2	1	0	0	1
	<code>canopy_cond_unstressed</code>	2	4	1	3	1
Turbulence	<code>humidity_scaling</code>	10	14	3	3	0
	<code>roughness</code>	3	3	1	0	0

The full JSBACH model in its ICON configuration is composed of about 30 different tasks and represents roughly 30'000 lines of code. There is a total of 70 `ELEMENTAL` subroutines and functions to transform. Our approach is applied to all required tasks for a climate simulation.


Table 1 describes the list of tasks automatically ported to GPU. For each tasks, the number of kernels generated and their corresponding inputs and outputs are detailed. 1 dimensional fields includes only the horizontal dimension where 2 dimensional include the number of soil layer as their second dimension.

## 2. The CLAW Compiler and the Single Column Abstraction

This section presents the CLAW Compiler and the Single Column Abstraction on which our work is based.

### 2.1. The Compiler

The CLAW Compiler is an extensible open-source source-to-source compiler for modern Fortran code. It is based on the OMNI Compiler Project [9].



**Figure 3.** CLAW Compiler workflow including the JSBACH DSL preprocessing

Figure 3 illustrates the internal workflow of the compiler. JSBACH code is pre-processed by a dedicated Python script which initially translates the JSBACH DSL to standard Fortran before entering the CLAW workflow. Fortran code is pre-processed and then parsed to the XcodeML/F Intermediate Representation (IR) [12]. This IR - represented as an Abstract Syntax Tree (AST) - is then manipulated by CLAW XcodeML to XcodeML Translator (X2T) to produce the refactored version of the code for a specific target with inserted directives. Finally, the IR is decompiled to standard Fortran code before being compiled by default compilers. Major contribution to extend its transformations for ELEMENTALS was introduced in the CLAW X2T.

### 2.2. CLAW Single Column Abstraction (SCA)

In weather and climate models, the physical parameterizations are, in most cases, single column processes. Each column of the three-dimensional computational domain can be computed

independently and does not have any data dependencies on its neighbors. The CLAW Single Column Abstraction DSL [1] was introduced to exploit this information. A CLAW SCA source code is fully standard Fortran code with the addition of specific CLAW directives. The code has no notion of the horizontal dimensions in the declaration of the fields as well as in the execution part. Loops iterating over horizontal dimension are omitted.

```
1 $ clawfc --target=gpu --directive=acc -o transformed.f90 original.f90
```

**Figure 4.** Invocation of the CLAW Compiler with a specified target architecture and desired directives

The SCA code is processed by the CLAW Compiler as shown in Fig. 4. The user defines the target architecture as well as the compiler directives to be used. Under the hood, the CLAW Compiler performs create a data dependency graph to decide where to create kernels and therefore which temporary fields have to be promoted. Based on this analysis it inserts the necessary iterations over the parallel dimensions and promotes the appropriate fields within the physical parametrization. In addition, the compiler inserts the required compiler directives to parallelize the loops and to manage data movement between the host and the device memory. The SCA transformation comes with various user-configurable options to manage promotion and data movement strategy, giving the end-user freedom to test various configurations.

### 3. Extension for ELEMENTAL Subroutines and Functions

An ELEMENTAL function or subroutine can be seen as a specific case of the CLAW SCA [1]. The land surface scheme is a point-wise computation on the grid that can be viewed as a column-wise computation with a limited number of vertical level.

The depth of the subroutine or function in the call graph determines the transformation applied to it. In Fig. 2, the leaf function `lwnet_from_lwdown` can be considered as a function to be inlined but `calc_radiation_surface_net` is a perfect candidate to apply SCA transformation rules.

Figure 5 is the original code of `calc_radiation_surface_net` processed by the user with CLAW SCA directive. Line 6 and 10 are defining the block of field coming from the model. These fields are often defined globally and therefore have to comply with the memory layout imposed by the calling model. The compiler then manages whether to apply promotion or not. The block directive also marks the whole subroutine as a SCA subroutine and thus activates the CLAW SCA transformation on it.

When a field is promoted or when a parallel iteration is inserted back into the code by the compiler, the dimension information is taken from the SCA model configuration file. This TOML formatted file defines the dimensions omitted in ELEMENTAL as well as layouts to be applied during the promotion transformation. This allows the user to investigate different data layouts depending on the target architecture. The user can update the default layout in the configuration file (Fig. 6) or add a layout clause to the `model-data` directive with the desired layout. The SCA model configuration file used for JSBACH is shown in Fig. 6. This file is read by the compiler before applying transformation.

The CLAW SCA transformation applied to non-ELEMENTAL subroutine or function will modifies the source code for both CPU and GPU targets. Since the original code with ELEMENTALS is already suited for CPU target, the extension of the SCA has no effect for this

```

1 ELEMENTAL SUBROUTINE calc_radiation_surface_net(swvis_down, swnir_down, &
2     alb_vis, alb_nir, lw_down, t, rad_net, swvis_net, swnir_net, sw_net, lw_net)
3
4     USE mo_phy_schemes, ONLY: lwnet_from_lwdown
5
6     !$claw model-data
7     REAL(wp), INTENT(in) :: swvis_down, swnir_down, alb_vis, alb_nir, lw_down, t
8     REAL(wp), INTENT(out) :: rad_net
9     REAL(wp), INTENT(out), OPTIONAL :: swvis_net, swnir_net, sw_net, lw_net
10    !$claw end model-data
11
12    REAL(wp) :: zswvis_net, zswnir_net, zsw_net, zlw_net
13
14    ! Compute net SW radiation from downward SW and albedo
15    zswvis_net = swvis_down * (1._wp - alb_vis)
16    zswnir_net = swnir_down * (1._wp - alb_nir)
17    zsw_net = zswvis_net + zswnir_net
18    ! Compute LW net radiation from incoming and the thermal radiation
19    zlw_net = lwnet_from_lwdown(lw_down, t)
20    ! Compute net radiation
21    rad_net = zsw_net + zlw_net
22
23    IF (PRESENT(swvis_net)) swvis_net = zswvis_net
24    IF (PRESENT(swnir_net)) swnir_net = zswnir_net
25    IF (PRESENT(sw_net)) sw_net = zsw_net
26    IF (PRESENT(lw_net)) lw_net = zlw_net
27 END SUBROUTINE calc_radiation_surface_net

```

**Figure 5.** Original source code for an ELEMENTAL subroutine enhanced with CLAW SCA directive inserted by the user

```

1 # CLAW SCA model configuration for JSBACH in ICON
2
3 [model]
4   name = "ICON_JSBACH"
5
6 [[dimensions]] # Definition of dimensions that can be used in layouts
7   id = "jsb_hori" # Horizontal dimension definition for JSBACH
8   [dimensions.size]
9     lower = 1 # if not specified, 1 by default
10    upper = "nc" # Number of columns
11
12 [[dimensions]]
13   id = "jsb_soil" # Dimension definition for the number of soil layers
14   [dimensions.size]
15     lower = 1 # Lower bound
16     upper = "nsoil" # Upper bound for the number of soil layers
17
18 [[layouts]] # Definition of layouts and default layout for specific target
19   id = "default" # mandatory layout, used if no specific target layout
20     # specified in the sca directive
21   position = [ "jsb_hori", ":" ] # Insert jsb_hori before the existing dimensions
22
23 [[layouts]]
24   id = "nc_nsoil" # Name of the layout
25   # Dimension layout for promotion. jsb_hori and jsb_soil are inserted before
26   # existing dimensions in the given order.
27   position = [ "jsb_hori", "jsb_soil", ":" ]

```

**Figure 6.** CLAW SCA configuration file for ICON model in TOML format

```

1 SUBROUTINE calc_radiation_surface_net ( swvis_down , swnir_down , alb_vis , &
2   alb_nir , lw_down , t , rad_net , swvis_net , swnir_net , sw_net , lw_net)
3
4   USE mo_phy_schemes , ONLY: lwnet_from_lwdown
5
6   REAL ( KIND = wp ) , INTENT(IN) :: swvis_down ( : )
7   REAL ( KIND = wp ) , INTENT(IN) :: swnir_down ( : )
8   REAL ( KIND = wp ) , INTENT(IN) :: alb_vis ( : )
9   REAL ( KIND = wp ) , INTENT(IN) :: alb_nir ( : )
10  REAL ( KIND = wp ) , INTENT(IN) :: lw_down ( : )
11  REAL ( KIND = wp ) , INTENT(IN) :: t ( : )
12  REAL ( KIND = wp ) , INTENT(OUT) :: rad_net ( : )
13  REAL ( KIND = wp ) , INTENT(OUT) :: swvis_net ( : )
14  REAL ( KIND = wp ) , INTENT(OUT) :: swnir_net ( : )
15  REAL ( KIND = wp ) , INTENT(OUT) :: sw_net ( : )
16  REAL ( KIND = wp ) , INTENT(OUT) :: lw_net ( : )
17  REAL ( KIND = wp ) :: zswvis_net
18  REAL ( KIND = wp ) :: zswnir_net
19  REAL ( KIND = wp ) :: zsw_net
20  REAL ( KIND = wp ) :: zlw_net
21  INTEGER :: jsb_hori
22
23  !$acc data &
24  !$acc present(swvis_down , swnir_down , alb_vis , alb_nir , lw_down , t , rad_net &
25  !$acc , swvis_net , swnir_net , sw_net , lw_net)
26  !$acc parallel
27  !$acc loop gang vector
28  DO jsb_hori = 1 , size ( swnir_down , 1 ) , 1
29    zswvis_net = swvis_down ( jsb_hori ) * ( 1._wp - alb_vis ( jsb_hori ) )
30    zswnir_net = swnir_down ( jsb_hori ) * ( 1._wp - alb_nir ( jsb_hori ) )
31    zsw_net = zswvis_net + zswnir_net
32    zlw_net = lwnet_from_lwdown ( lw_down ( jsb_hori ) , t ( jsb_hori ) )
33    rad_net ( jsb_hori ) = zsw_net + zlw_net
34    IF ( present ( swvis_net ) ) THEN
35      swvis_net ( jsb_hori ) = zswvis_net
36    END IF
37    IF ( present ( swnir_net ) ) THEN
38      swnir_net ( jsb_hori ) = zswnir_net
39    END IF
40    IF ( present ( sw_net ) ) THEN
41      sw_net ( jsb_hori ) = zsw_net
42    END IF
43    IF ( present ( lw_net ) ) THEN
44      lw_net ( jsb_hori ) = zlw_net
45    END IF
46  END DO
47  !$acc end parallel
48  !$acc end data
49 END SUBROUTINE calc_radiation_surface_net

```

Figure 7. Transformed code with OpenACC directives

target in this case. We rely here on the Fortran compiler to efficiently compile ELEMENTAL on CPU.

When applied to GPU, the following actions occur for non-leaf subroutine or function:

- The signature of the subroutine/function is updated and the ELEMENTAL or PURE specifiers are discarded.
- The flagged fields are promoted according to the specified layout. If no layout is specified, the default layout is assumed. The promotion and layout information come from the configuration file as shown in Fig. 6.
- Iterations over new dimensions specified in the layout are inserted.
- Data analysis is performed and temporary fields or scalars that need promotion are promoted.

Leaf ELEMENTAL subroutines and functions have simpler transformations applied to them.

The code is processed as follows:



- As for other ELEMENTAL subroutines/functions, the signature is updated and the ELEMENTAL or PURE specifiers are discarded.
- Compiler directives are inserted to set the subroutine or function as an OpenACC routine or OpenMP target code.

Once the code is annotated with CLAW SCA directives, the CLAW Compiler is called the same way for file with SCA or SCA with ELEMENTALS. Listing 4 can be executed in the same way.

### 3.1. Expansion Directive

As mentioned in Section 1, vector notation is used widely in tasks. To automatize the port of these blocks of Fortran code to OpenACC and OpenMP, the CLAW expand directive is used.

```

1 !$claw expand parallel
2   rad_srf_net (:) = (1._wp - fract_lice (:)) * rad_net_lwtr (:) + fract_lice (:) *
   rad_net_lice (:)
3   sw_srf_net (:) = (1._wp - fract_lice (:)) * sw_net_lwtr (:) + fract_lice (:) *
   sw_net_lice (:)
4   lw_srf_net (:) = (1._wp - fract_lice (:)) * lw_net_lwtr (:) + fract_lice (:) *
   lw_net_lice (:)
5   !$claw end expand

```

**Figure 8.** Block of vector notation with CLAW expand directive

```

1 !$acc parallel loop gang vector DO claw_induction_0 = 1 , size ( rad_srf_net , 1
   )
2   rad_srf_net ( claw_induction_0 ) = ( 1._wp - fract_lice ( claw_induction_0 &
   ) ) * rad_net_lwtr ( claw_induction_0 ) + fract_lice ( claw_induction_0 ) &
3   * rad_net_lice ( claw_induction_0 )
4   sw_srf_net ( claw_induction_0 ) = ( 1._wp - fract_lice ( claw_induction_0 &
   ) ) * sw_net_lwtr ( claw_induction_0 ) + fract_lice ( claw_induction_0 ) &
5   * sw_net_lice ( claw_induction_0 )
6   lw_srf_net ( claw_induction_0 ) = ( 1._wp - fract_lice ( claw_induction_0 &
   ) ) * lw_net_lwtr ( claw_induction_0 ) + fract_lice ( claw_induction_0 ) &
7   * lw_net_lice ( claw_induction_0 ) END DO
8
9
10

```

**Figure 9.** Expanded vector notation

Figure 8 presents a block of vector notation operations surrounded by a CLAW expand block. The expand block works only for specific target as the CLAW SCA for ELEMENTAL. Before the transformation is applied to the block, an analysis is performed to make sure all the array dimensions are compatible to be transformed within the same loop structure. If the criteria are met, iteration are inserted on appropriate range and parallelization for the given directives. Figure 9 is the same code after transformation applied to it.

## 4. Performance Comparison

In this section we present the performance results that have been achieved by applying the extended version of the CLAW Compiler to the JSBACH land surface scheme. The computational domain size (number of horizontal grid points  $\times$  number of soil layers) used for the performance measurement is 20480 $\times$ 5. All performance measurements in this section are obtained using -O3 optimization flag or equivalent for each compiler. The code is transformed with the CLAW Compiler 2.0 and compiled with the PGI 18.10 Fortran compiler for the baseline

CPU as well as the GPU OpenACC results. The CPU reference is obtained using multi-core OpenMP parallelism available in the original version of ICON. The GPU OpenMP target results were obtained with Cray Compiler CCE 8.7.4 and by running standalone version of the kernels since the full ICON model is not ready to have a code mixing OpenMP for multi-core and for accelerator.

Figure 10 shows the speedup achieved from the CLAW SCA transformed version with OpenACC directives and the CLAW SCA transformed version with OpenMP directives over the CPU reference. The original version of the code is exploiting the 12 cores available on the Intel Xeon E5-2690 v3 Haswell CPU of Piz Daint at CSCS parallelized with multi-core OpenMP. The CLAW OpenACC and OpenMP versions are executed on one NVIDIA P100 GPU. The theoretical floating point peak performance of the Intel Haswell is 500 GFLOPS, while the NVIDIA P100 is 5.3 TFLOPS. In term of memory bandwidth, the Haswell has a theoretical peak at 68 GB/s while the P100 can reach up to 732 GB/s. For both compute and memory bandwidth limited problem one can expect a maximum speedup of 11x.

As shown, OpenACC and OpenMP results are very similar and expected. Kernels issued from the ELEMENTAL transformation are pretty simple to be handled by a compiler and we did not expect PGI and Cray to fundamentally generate different code for them. Depending on the size of the kernel and the tile it has to process, we are able to achieve speedup between 1.7x up to 8.6x for kernel like the `albedo: calc_sky_view_fractions`. Tiles can be viewed as a collection of grid points from a specific type of land (e.g., lake or glacier) or as a mask for this specific type of land. Therefore tiles can have more or less work to be performed due to a different set of grid points. As GPUs need enough work to exploit massive parallelism, a reduced set of grid points is one of the reason the speedup can vary this much. Another factor is the size of the kernel. Some ELEMENTAL functions/subroutines are very small and the kernels generated from them are also small. Kernel's launch and synchronization is then a non-negligible part of the overall kernel execution time. These overhead can be hidden in the execution of the full model with asynchronous mechanism.


## 5. Code Metrics

This section briefly compares the original code with its CLAW SCA version before and after applying transformations. Unlike the original SCA the SCA for ELEMENTAL imposes less changes to the code. In order to comply with the specification, the user only annotates the ELEMENTAL subroutines and functions to be executed on GPU, and relies on the compiler from then on. There is no need to apply other code change in the ELEMENTALS.

The full JSBACH model is annotated with 40 CLAW SCA directives and about 130 CLAW expansion directives to convert vector notation to parallelizable loops. In the transformed code there are almost 2000 lines of OpenACC directives or OpenMP depending the user's choice. This is the amount of compiler directives a user would have had to write by hand to achieve the same goal. A hand-written version would also have to change the code structure significantly, for example by promoting fields and adding loops.

## Conclusion

In this paper we propose an extension of the CLAW SCA DSL and its compiler to take advantage of the ELEMENTAL construct in Fortran code and automatize the port to heteroge-



**Figure 10.** Performance comparison (socket to socket) between CLAW OpenACC, CLAW OpenMP and the CPU reference of kernels from two JSBACH tasks on Intel Haswell E5-2690v3 and NVIDIA P100. Domain size (number of horizontal grid points  $\times$  number of soil layers) =  $20480 \times 5$

nous architecture. The ELEMENTAL construct as exploited in JSBACH provides the necessary abstraction to implement automatic code transformation and target GPU architectures without writing lots of compiler directives by hand. Promotion, loop parallelization and compiler directive generation are handled by the CLAW Compiler automatically.

For this paper, the CLAW SCA extension was applied to a wide portion of the JSBACH land surface scheme, one of the physical parametrizations of the ICON global climate model. From a single simple source code multiple programming paradigm can be targeted. Performance

results show up to 8.6x speedup against a 12-core parallel CPU version for a specific kernel. All kernels are at least 1.7x faster than the CPU version. The overall performance of the JSBACH model running on GPU is typically between 5x to 6x speedup depending on its configuration.

In the current implementation of the extension, a single parallel subroutine or function is generated from its ELEMENTAL counterpart. This is fine for the JSBACH use case as a single ELEMENTAL is always called with the same kind of arguments. Future improvement to generate several versions of a subroutine or a function if the type of argument used to call them is different.

As it is possible to generate any source code, we can imagine to take advantage of new compiler development such as exploiting the `DO CONCURRENT` construct from Fortran 2008 to target accelerators as it is investigated in latest version of PGI. Instead of generating compiler directives the CLAW Compiler could exploit this new Fortran feature.

## Acknowledgements

This work was supported by a grant from the Swiss National Supercomputing Centre (CSCS) under project ID c15. We would like to show our gratitude to the OMNI Compiler Team at RIKEN CCS for their effort and support to provide a very useful open-source tool-chain. We would also like to thank Reiner Schnur, a scientific programmer at the Max-Planck-Institut für Meteorologie for his insights in the JSBACH code base and review of our changes. This work was partly funded by ETH Zurich and the PASC Initiative under the ENIAC project.

*This paper is distributed under the terms of the Creative Commons Attribution-Non Commercial 3.0 License which permits non-commercial use, reproduction and distribution of the work without further permission provided the original work is properly cited.*

## References

1. Clement, V., Ferrachat, S., Fuhrer, O., et al.: The CLAW DSL: Abstractions for performance portable weather and climate models. In: Proceedings of the Platform for Advanced Scientific Computing Conference. pp. 2:1–2:10. PASC '18, ACM, New York, NY, USA (2018), DOI: 10.1145/3218176.3218226
2. Crueger, T., Giorgetta, M.A., Brokopf, R., et al.: ICON-A, the atmosphere component of the ICON Earth system model: II. model evaluation. Journal of Advances in Modeling Earth Systems 10(7), 1638–1662 (2018), DOI: 10.1029/2017MS001233
3. Fuhrer, O., Osuna, C., Lapillonne, X., et al.: Towards a performance portable, architecture agnostic implementation strategy for weather and climate models. Supercomputing Frontiers and Innovations 1(1), 45–62 (2014), DOI: 10.14529/jsfi140103
4. Giorgetta, M.A., Brokopf, R., Crueger, T., et al.: ICON-A, the atmosphere component of the ICON Earth system model: I. model description. Journal of Advances in Modeling Earth Systems 10(7), 1613–1637 (2018), DOI: 10.1029/2017MS001242
5. Gysi, T., Osuna, C., Fuhrer, O., et al.: Stella: A domain-specific tool for structured grid methods in weather and climate models. In: Proceedings of the International Conference for

- High Performance Computing, Networking, Storage and Analysis. pp. 41:1–41:12. SC '15, ACM, New York, NY, USA (2015), DOI: 10.1145/2807591.2807627
6. Lapillonne, X., Fuhrer, O.: Using compiler directives to port large scientific applications to GPUs: An example from atmospheric science. *Parallel Processing Letters* 24(1) (2014), DOI: 10.1142/S0129626414500030
  7. Mauritsen, T., Bader, J., Becker, T., et al.: Developments in the MPI-M Earth System Model version 1.2 (MPI-ESM1.2) and Its Response to Increasing CO<sub>2</sub>. *Journal of Advances in Modeling Earth Systems* 11(4), 998–1038 (2019), DOI: 10.1029/2018MS001400
  8. Muller, M., Aoki, T.: Hybrid Fortran: High productivity GPU porting framework applied to Japanese weather prediction model. *CoRR* abs/1710.08616 (2017), <http://arxiv.org/abs/1710.08616>
  9. Omni CompilerProject: Omni Compiler Project - An Infrastructure for Source-to-Source Transformation. <http://omni-compiler.org> (2013-2019), accessed: 2019-09-02
  10. OpenACC Standard: The OpenACC application programming interface - version 2.7. <https://www.openacc.org/sites/default/files/inline-files/OpenACC.2.7.pdf> (2018), accessed: 2019-09-02
  11. OpenMP Architecture Review Board: OpenMP application programming interface - version 5.0. <https://www.openmp.org/wp-content/uploads/OpenMP-API-Specification-5.0.pdf> (2018), accessed: 2019-09-03
  12. XcalableMP Specification Working Group: XcodeML/Fortran Specification. <https://omni-compiler.org/download/xcodeml/stable/XcodeML-F-1.0.pdf> (2017), accessed: 2019-09-02